デザインパターン勉強会①Iterator

<h2>はじめに</h2>
ZOOM勉強会後、覚え書きとして、内容をまとめてみました。
初心者二人での勉強会のため、間違っている点などあるかもしれません。

<h2>勉強会について</h2>
<h3>目的</h3><ul><li>学習を習慣化する</li><li>オブジェクト指向・デザインパターンへの理解を深める</li><li>業務に活かす</li></ul>

<h3>内容</h3>結城 浩さん著の<a href="https://www.amazon.co.jp/%E5%A2%97%E8%A3%9C%E6%94%B9%E8%A8%82%E7%89%88Java%E8%A8%80%E8%AA%9E%E3%81%A7%E5%AD%A6%E3%81%B6%E3%83%87%E3%82%B6%E3%82%A4%E3%83%B3%E3%83%91%E3%82%BF%E3%83%BC%E3%83%B3%E5%85%A5%E9%96%80-%E7%B5%90%E5%9F%8E-%E6%B5%A9/dp/4797327030"><b>「増補改訂版Java言語で学ぶデザインパターン入門」</b></a>を読み、プログラムを作成して、デザインパターンを理解します。<br>

初回は<b>Iterator</b>パターンについて学びました。


<h3>環境</h3>
VB.NET
Visual Studio 2017 & 2019 
.NET Framework 4.7.2

<h2>Iteratorパターン</h2>
Iteratorパターンとは数え上げを行うパターンです。

登場する役割は以下のとおりです。
<ul><li>Iterator</li>要素を順番にスキャンしていく反復子のインタフェース<li>ConcreteIterator</li>Iteratorを具体的に実装<li>Aggregate</li>Iterator役を作り出すインタフェース<li>ConcreteAggregate</li>Aggregateが定めたインタフェースを実装</ul>

<h3>Iterator</h3>
.NetFrameworkではIterator役として`IEnumerator`が用意されています。
各メソッドの役割は、コメントに記述しているとおりです。

```vb:IEnumerator
Public Interface IEnumerator
    '列挙子の現在位置にあるコレクション内の要素を取得します。
    ReadOnly Property Current As Object

    '列挙子を初期位置、つまりコレクションの最初の要素の前に設定します。
    Sub Reset()

    '列挙子をコレクションの次の要素に進めます。
    '次の要素に正常に進んだ場合Trueを返します。
    Function MoveNext() As Boolean
End Interface
```

<h3>Aggregate</h3>
一方、Aggregate役として`IEnumerable`が用意されています。
Aggregate役は、Iterator役を返すメソッドだけを定義しています。

```vb:IEnumerable
Public Interface IEnumerable
    'コレクションを反復処理する列挙子を返します。
    Function GetEnumerator() As IEnumerator
End Interface
```
今回はこれらのIEnumeratorとIEnumerableを使用してIteratorを実装してみました。


<h3>ConcreteAggregate</h3>
以下のMyListクラスが今回のConcreteAggregate役です。


```vb:MyList
Public Class MyList : Implements IEnumerable

    ' 要素の集合
    Private _elements As Object()

    ' 要素数
    Private _last As Integer = 0

    Public Sub New(ByVal maxSize As Integer)
        ReDim _elements(maxSize)
    End Sub

    ' indexの要素を取得する
    Public Function GetElementAt(ByVal index As Integer) As Object
        Return _elements(index)
    End Function

    ' 要素を追加する
    Public Sub AppendElement(ByVal obj As Object)
        _elements(_last) = obj
        _last += 1
    End Sub

    ' 要素の数を取得する
    Public Function GetLength() As Integer
        Return _last
    End Function

    ' 反復処理する列挙子を返す
    Public Function GetEnumerator() As IEnumerator Implements IEnumerable.GetEnumerator
        Return New MyListIterator(Me)
    End Function


End Class
```
`MyList`は、Aggregate役の`IEnumerable`を実装しており、`GetEnumerator()`でIterator役の`IEnumerator`を返します。
このメソッドは、Iteratorパターンを使用する側のコードから呼び出します（Mainなど）。

また、このクラスは各要素と要素数の取得ができるようになっています。
次に説明するConcreteIteratorでは、これらのメソッドを使用して集合体の情報にアクセスします。



<h3>ConcreteIterator</h3>
以下の`MyListIterator`が今回のConcreteIterator役です。



```vb:MyListIterator
Public Class MyListIterator : Implements IEnumerator

    '集合体
    Private _myList As MyList

    '今返しているインデックス
    Private _index As Integer

    Public Sub New(ByVal myList As MyList)
        Me._myList = myList
        _index = -1
    End Sub

    '現在の要素を返す
    Public ReadOnly Property Current As Object Implements IEnumerator.Current
        Get
            Dim obj As Object = _myList.GetElementAt(_index)
            Return obj
        End Get
    End Property

    'インデックスをリセット
    Public Sub Reset() Implements IEnumerator.Reset
        _index = -1
    End Sub

    '次のインデックスにデータが存在するかを返す
    Public Function MoveNext() As Boolean Implements IEnumerator.MoveNext
        _index +=1
        If _index < _myList.GetLength Then
            Return True
        Else
            Return False
        End If
    End Function
End Class
```
`MyListIterator`は、Iterator役の`IEnumerator`を実装しています。
このクラスは、コンストラクタでConcreteAggregate役の`MyList`を渡されています。
ここで渡されている集合体に対して、数え上げを行うことになります。

各メソッドについて説明します。

`Current`プロパティでは、`Get`で現在のインデックスが指し示す要素を返しています。

`MoveNext()`では、次の要素に進めることができるかを返しています。
以下のコードでは、単純にインデックスを一つだけ進めていますが、ここの実装を変えることで、様々な数え上げを表現することができます。

`Reset()`では、インデックスを初期値に戻します。
Java言語で学ぶデザインパターン入門のIteratorには、このメソッドはありませんでした。
勉強会では、Iteratorは使い捨てでよく、`Reset()`で使い回すのは良くないのでは？という意見もありました。


<h3>使い方</h3>
以上で、すべての役が出揃いました。
実際に数え上げを行うコードは以下です。

```vb:Main
Sub Main()
    '準備
    Dim mylist As New MyList(10)
    mylist.AppendElement("a")
    mylist.AppendElement("b")
    mylist.AppendElement("c")
    mylist.AppendElement(1)

    'ここからがIteratorパターン活用部分    
    Dim it As IEnumerator = mylist.GetEnumerator()

    While it.MoveNext()
        Debug.Write(it.Current)
    End While
    '出力結果↓
    'abc1

    'これが上と同じことをしている
    For Each element In mylist
        Debug.Write(element)
    Next
    '出力結果↓
    'abc1
End Sub
```
`GetEnumerator()`でIteratorを取得し、そのIteratorの`MoveNext()`で要素を進め、`Current`で要素を取得します。
`For Each`文は、このIteratorパターンの糖衣構文だそうです。

Iteratorパターンで数え上げを行っている部分を見ると、`MyList`に実装されているメソッドは出てきていません。
つまりWhileループは、`MyList`には依存しないことになり、`MyList`の中身が配列であろうとListであろうと数え上げを行うことができます。
`MyList`がIteratorを返すことができれば、`MyList`を修正しても、数え上げの部分を修正しなくて良くなります。
以上がIteratorパターンの説明になります。


<h2>ついでに勉強したこと</h2>

<h3>Iterator Function</h3>

VB.NETでは、`IEnumerable`型を返す`Iterator Function`というものがあるようです。
この`Iterator Function`を定義することで、簡易的な数え上げの機構をつくることができます。
以下が`Iterator Function`のサンプルコードです。


```vb:MyList2
Public Class MyList2

    Private _elements As Object()
    Private _last As Integer = 0 '要素数

    '
    '  addなどは省略
    '

    '例1
    Iterator Function GetEnumerable() As IEnumerable(Of Object)
        For i = 0 To _last - 1
            Yield _elements(i)
        Next
    End Function

    '例2 こんなふうにも書けます
    Iterator Function GetEnumerable2() As IEnumerable(Of Object)
        Yield 1
        Yield 2
        Yield 3
    End Function

End Class
```
`Iterator Function`の中では、`Yield`が来るとその値を返し、一旦この関数を抜けます。
呼び出し側の処理が終わり次の要素を要求されたとき、先程の続きから処理を開始し、次の`Yield`でまた関数を抜け値を返します。
これを繰り返して数え上げを行います。
1つ目の例だと、単純に配列の要素を順番に返しています。
2つ目の例だと、1,2,3を順番に返しています。


呼び出し側は、以下の様になっています。

```vb:IteratorFunctionMain
    For Each element In mylist2.GetEnumerable
        Debug.WriteLine(element)
    Next
```

`For Each`の部分で`Iterator Function`が呼ばれます。
`Iterator Function`では、`Yield`が来るたびに一旦処理を抜け、返された値が`element`に入ります。
`Next`まで行くと、処理を抜けた`Iterator Function`の途中から処理が始まり、次の`Yield`で、次の`element`が決まります。
先程の1つ目の例だと、配列の要素が順番に入ります。
2つ目の例だと、1,2,3が順番に入ります。
これを繰り返し、数え上げが行われています。





<h2>まとめ</h2>
今回はIteratorパターンについて学びました。
VB.NETでは、IEnumerableとIEnumeratorがあり、普段Listなどを使うときにはIteratorの恩恵を受けていたようです。

勉強会では開発に時間をかけるほどお金がかかるという意見もあり、配列が出てくるたびにIteratorパターンを実装するのがよいわけではないと思います。
今後自分たちが実装する機会は少ないかもしれませんが、IEnumerableとIEnumeratorが何をしているか理解でき良い勉強になりました。